Direction Fields and Solution Curves

We begin by re-initializing the Maple "kernel" and loading the additional packages that we are most likely to use, the plots package and the DEtools package. The ***plots package*** contains extra plotting commands including fieldplot, implicitplot, and odeplot. The ***DEtools package*** contains various graphical commands for studying a differential equation numerically. We use it in this worksheet to draw direction fields and to plot solution curves.

```
> restart;
> with( plots );
```

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *dualaxisplot*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *shadebetween*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

```
> with (DEtools);
```

[*AreSimilar*, *Closure*, *DEnormal*, *DEplot*, *DEplot3d*, *DEplot_polygon*, *DFactor*, *DFactorLCLM*, *DFactorsols*, *Dchangevar*, *Desingularize*, *FunctionDecomposition*, *GCRD*, *Gosper*, *Heunsols*, *Homomorphisms*, *IVPsol*, *IsHyperexponential*, *LCLM*, *MeijerGsols*, *MultiplicativeDecomposition*, *ODEInvariants*, *PDEchangecoords*, *PolynomialNormalForm*, *RationalCanonicalForm*, *ReduceHyperexp*, *RiemannPsols*, *Xchange*, *Xcommutator*, *Xgauge*, *Zeilberger*, *abelsol*, *adjoint*, *autonomous*, *bernoullisol*, *buildsol*, *buildsym*, *canoni*, *caseplot*, *casesplit*, *checkrank*, *chinisol*, *clairautsol*, *constcoeffsols*, *convertAlg*, *convertsys*, *dalembertsol*, *dcoeffs*, *de2diffop*, *dfieldplot*, *diff_table*, *diffop2de*, *dperiodic_sols*, *dpolyform*, *dsubs*, *eigenring*, *endomorphism_charpoly*, *equinv*, *eta_k*, *eulersols*, *exactsol*, *expsols*, *exterior_power*, *firint*, *firtest*, *formal_sol*, *gen_exp*, *generate_ic*, *genhomosol*, *gensys*, *hamilton_eqs*, *hypergeomsols*, *hyperode*, *indicialeq*, *infgen*, *initialdata*, *integrate_sols*, *intfactor*, *invariants*, *kovacicsols*, *leftdivision*, *liesol*, *line_int*, *linearsol*, *matrixDE*, *matrix_riccati*, *maxdimsystems*, *moser_reduce*, *muchange*, *mult*, *mutest*, *newton_polygon*, *normalG2*, *ode_int_y*, *ode_y1*, *odeadvisor*, *odepde*, *parametricsol*, *particularsol*, *phaseportrait*, *poincare*, *polysols*, *power_equivalent*, *rational_equivalent*, *ratsols*, *redode*, *reduceOrder*, *reduce_order*, *regular_parts*, *regularsp*, *remove_RootOf*, *riccati_system*, *riccatisol*, *rifread*, *rifsimp*, *rightdivision*, *rtaylor*, *separablesol*, *singularities*, *solve_group*, *super_reduce*, *symgen*, *symmetric_power*, *symmetric_product*, *symtest*, *transinv*, *translate*, *untranslate*, *varparam*, *zoom*]

A Qualitative Approach to a DE

In a **qualitative appoach**, we view a DE $\frac{d}{dx} y = f(x, y)$ geometrically. We do this by drawing a **direction field**, which is obtained by drawing through each point in the (x, y)-plane a short line segment or arrow with the slope $f(x, y)$. **Solutions**, or **integral curves**, of the DE have the property that at each of their points they are tangent to the direction field at that point, and thus the general **qualitative** nature of the solutions can be determined from the direction field.

Example 1.

As a first example, consider the DE, called **deq1**,

$$\frac{d}{dx} y = e^{-x} - 2y.$$

We use the DEplot command from the ***DEtools package*** to draw a direction field of **deq1**. Each arrow shows the direction of movement from various points on the plane.

```
> deq1:=diff(y(x),x)=exp(-x)-2*y(x);
```

$$deq1 := \frac{d}{dx} y(x) = e^{-x} - 2\,y(x)$$

```
> DEplot (deq1, y(x), x=-2..5, y=-1..2, arrows=medium, color=magenta)
  ;
```



The direction field strongly suggests that if a solution is negative at some point, then it is increasing at that point, and all solutions approach 0 as $x \to \infty$.

We find the general solution of **deq1**.

```
> dsolve(deq1,y(x));
```

$$y(x) = \left(e^{x} + \_C1\right) e^{-2x}$$

Now let's add 7 different initial conditions and calculate the corresponding specific solutions.

```
> for n from -1 by .5 to 2 do
  ic[n]:=y(0)=n;
  sol[n]:=dsolve({deq1,ic[n]},y(x));
  od;
```

$$ic_{-1} := y(0) = -1$$

$$sol_{-1} := y(x) = \left(e^{x} - 2\right) e^{-2x}$$

$$ic_{-0.5} := y(0) = -0.5$$

$$sol_{-0.5} := y(x) = \left(e^{x} - \frac{3}{2}\right) e^{-2x}$$

$$ic_{0.} := y(0) = 0.$$

$$sol_{0.} := y(x) = \left(e^{x} - 1\right) e^{-2x}$$

$$ic_{0.5} := y(0) = 0.5$$

$$sol_{0.5} := y(x) = \left(e^{x} - \frac{1}{2}\right) e^{-2x}$$

$$ic_{1.0} := y(0) = 1.0$$

$$sol_{1.0} := y(x) = e^{-2x} e^{x}$$

$$ic_{1.5} := y(0) = 1.5$$

$$sol_{1.5} := y(x) = \left(e^{x} + \frac{1}{2}\right) e^{-2x}$$

$$ic_{2.0} := y(0) = 2.0$$

$$sol_{2.0} := y(x) = \left(e^{x} + 1\right) e^{-2x}$$

We add these solutions to our direction field.

```
> p1:=DEplot (deq1, y(x), x=-2..5, y=-1..2,[seq([0,i],i=-1..2,.5)],
  arrows=medium,color=magenta,linecolor=blue):
  display(p1);
```



It turns out that each of the given solutions that start out negative have their maximum value along the curve

$$y = \frac{1\,\mathrm{e}^{-x}}{2}.$$

This is the red curve below. Thus all the solutions approach the x-axis from above.

```
> p2:=plot(1/2*exp(-x),x=-2..5,color=red):
> display(p1,p2);
```



Although it is hard to tell on the right, none of the solutions intersect. This is the result of the important following **existence and uniqueness theorem**.

Suppose that f and $\frac{\partial}{\partial y} f(x, y)$ are continuous functions in the rectangle $R=\{(x, y): a \leq x \leq b, c \leq y \leq d\}$
,
and that the point (x_0, y_0) lies in R. Then the intial value problem $\frac{d}{dx} y = f(x, y), y(x_0) = y_0$ has a unique solution $y(x)$ that exists at least as long as its graph stays within R.

Graphically, this theorem says that there is a smooth solution curve (or integral curve) through every point in R and that the solution curves cannot cross.

Example 2

We now consider a second DE, named **deq2**,

$$\frac{d}{dx} y = y^2 + x.$$

```
> deq2:=diff(y(x),x)=y(x)^2+x;
```

$$deq2 := \frac{d}{dx}\, y(x) = y(x)^2 + x$$

We plot its direction field.

```
> DEplot (deq2, y(x), x=-2..2, y=-2..2,arrows=medium,color=magenta);
```



From the plot, it appears that all solutions are increasing for x>0 and that all solutions eventually become positive. The graph also suggests all solutions apprach infinity, but does this happen at a finite value of x or only as $x \to \infty$. Let's find the solution.

```
> soln:=dsolve(deq2,y(x));
```

$$soln := y(x) = \frac{\_C1\ \mathrm{AiryAi}(1, -x) + \mathrm{AiryBi}(1, -x)}{\_C1\ \mathrm{AiryAi}(-x) + \mathrm{AiryBi}(-x)}$$

What, pray tell, is an AiryAi or AiryBi? Let's add an initial condition and find a specific solution.

```
> ic2:=y(0)=-1;
```

$$ic2 := y(0) = -1$$

```
> dsolve({deq2,ic2},y(x));
```

$$y(x) = \frac{\dfrac{\left(2\,3^{5/6}\,\pi + 3\,\Gamma\left(\frac{2}{3}\right)^2 3^{2/3}\right) \mathrm{AiryAi}(1, -x)}{3\,3^{1/6}\,\Gamma\left(\frac{2}{3}\right)^2 - 2\,\pi\,3^{1/3}} + \mathrm{AiryBi}(1, -x)}{\dfrac{\left(2\,3^{5/6}\,\pi + 3\,\Gamma\left(\frac{2}{3}\right)^2 3^{2/3}\right) \mathrm{AiryAi}(-x)}{3\,3^{1/6}\,\Gamma\left(\frac{2}{3}\right)^2 - 2\,\pi\,3^{1/3}} + \mathrm{AiryBi}(-x)}$$

We use evalf to find a numerical approximation.

```
> evalf(%);
```

$$y(x) = \frac{-11.05113710\,\mathrm{AiryAi}(1, -1.\,x) + \mathrm{AiryBi}(1, -1.\,x)}{-11.05113710\,\mathrm{AiryAi}(-1.\,x) + \mathrm{AiryBi}(-1.\,x)}$$

Let's evaluate this expression at $x = 2$.

```
> subs(x=2,%);
```

$$y(2) = \frac{-11.05113710\,\mathrm{AiryAi}(1, -2.) + \mathrm{AiryBi}(1, -2.)}{-11.05113710\,\mathrm{AiryAi}(-2.) + \mathrm{AiryBi}(-2.)}$$

Still not much help for us. But Maple will use a numeric method, as we will discuss in Secionn 1.4, to plot various solutions for us in **DEplot**.

```
> DEplot (deq2, y(x), x=-2..2, y=-2..2,[seq([0,i],i=-2..2,.5)],
  arrows=medium,color=magenta, linecolor=blue);
```



It turns out that all the solutions go to infinity in finite time.

Next we consider the differential equation $\frac{d}{dx}\, y = -\frac{y}{x}$.

```
> deq3:=diff(y(x),x)=-y(x)/x;
```

$$deq3 := \frac{d}{dx}\, y(x) = -\frac{y(x)}{x}$$

```
> DEplot (deq3, y(x), x=-3..3, y=-3..3, arrows=medium, color=magenta)
  ;
```



solutions are decreasing in quadrants I and III, increasing in quadrants II and IV. Also, solutions are concave up in quadrants I and II, concave down in quadrants III and IV. Solutions in quadrants I and IV approach 0 as $x \to \infty$. Solutions in quadrant II approach ∞ as $x \to 0$ and soluions in quadrant III approach $-\infty$ as $x \to 0$.

We find the general solution of **deq3**.

```
> dsolve(deq3,y(x));
```

$$y(x) = \frac{\_C1}{x}$$

Now let's add some initial conditions and calculate the corresponding specific solutions.

```
> ic[1]:=y(-1)=2;
  sol[1]:=dsolve({deq3,ic[1]},y(x));
```

$$ic_1 := y(-1) = 2$$

$$sol_1 := y(x) = -\frac{2}{x}$$

```
> ic[2]:=y(3)=1;
  sol[1]:=dsolve({deq3,ic[2]},y(x));
```

$$ic_2 := y(3) = 1$$

$$sol_1 := y(x) = \frac{3}{x}$$

```
> ic[3]:=y(1)=-1;
  sol[1]:=dsolve({deq3,ic[3]},y(x));
```

$$ic_3 := y(1) = -1$$

$$sol_1 := y(x) = -\frac{1}{x}$$

```
> ic[4]:=y(-1)=-2;
  sol[1]:=dsolve({deq3,ic[4]},y(x));
```

$$ic_4 := y(-1) = -2$$

$$sol_1 := y(x) = \frac{2}{x}$$

We add these solutions to our direction field.

```
> p1:=DEplot (deq3, y(x), x=-4..4, y=-4..4,[[-1,2],[3,1],[1,-1],[-1,
  -2]],arrows=medium,color=magenta,linecolor=blue):
  display(p1);
```



No initial condition (well almost) can have a first coordinate of 0 since $f(x, y) = -\frac{y}{x}$ is not defined and thus not continuous at any (0,y) except (0,0) where the solution is $y = 0$. But Theorem 1 applies if an initial condition is not on the y-axis and the Rectangle R does not include any points on the y-axis.

Page 22 # 6

Consider the DE $\frac{d}{dx} y = x + \sin y$.

```
> deq:=diff(y(x),x)=x+sin(y(x));
```

$$deq := \frac{d}{dx} y(x) = x + \sin(y(x))$$

(a) A solution curve passes through the point $\left(1, \frac{\pi}{2}\right)$. What is the slope at this point?

```
> slope:=1+sin(Pi/2);
```

$$slope := 2$$

(b) Argue that every solution curve is increasing for $x > 1$. The solution curve is increasing where

$\frac{d}{dx} y > 0.$

```
> solve(x+sin(y)>0,x);
```

$$\{-\sin(y) < x\}$$

Since $-1 \leq \sin(y) \leq 1$, $-1 \leq -\sin(y) \leq 1$. Thus $-\sin(y) < x$ for $x > 1$. We look at the solution of the above inequality graphically.

```
> inequal(x+sin(y)>0,x=-10..10,y=-10..10);
```



We look at the direction field.

```
> DEplot (deq, y(x), x=-4..4, y=-4..4,arrows=medium,color=magenta);
```



(c) Show that the second derivative of every solution satisfies $\frac{d^2y}{dx^2} = 1 + x cos(y) + \frac{1}{2} \cdot \sin(2 \cdot y)$.

```
> yprime:=x+sin(y);
```

$$yprime := x + \sin(y)$$

```
> y2prime:=1+cos(y)*yprime;
```

$$y2prime := 1 + \cos(y)\ (x + \sin(y))$$

```
> y2prime:=expand(y2prime);
```

$$y2prime := 1 + \cos(y)\ x + \cos(y)\ \sin(y)$$

Since $\sin(2 \cdot y) = 2 \sin(y) \cos(y)$, $y2prime = 1 + x \cdot \cos(y) + \frac{1}{2} \sin(2 \cdot y)$.

(d) A solution curve passes through $(0, 0)$. Prove that this curve has a relative minimum at $(0, 0)$.

```
> eval(yprime,[x=0,y=0]);
```

$$0$$

```
> eval(y2prime,[x=0,y=0]);
```

$$1$$

Since $(0, 0)$ is a critical point and is concave up there, we have a local minimum at $(0, 0)$. We add the graph of the solution to the direction field.

```
> DEplot (deq, y(x), x=-5..5, y=-1..10,[[0,0]],arrows=medium,color=
  magenta,linecolor=blue);
```